Tools for Comparing Expressions

Maple 2025 provides innovative new tools for comparing expressions.

ExpressionTools

- Have you ever tried to compare two large expressions but you couldn't easily see how they differed? For example:

```
> e1 := 1/4*(2*T[x,j,g]*T[i,c,k]-2*T[x,j,c]*T[i,g,k]-2*d[j](T[x,c,i],[X])+2*d[x](T[c,j,i],[X]))*g_[a,b]+1/4*(T[a,x,b]+T[b,x,a])*T[c,j,i]-1/4*T[b,x,j]*T[c,a,i]-1/2*T[x,j,a]*T[i,b,c]+1/2*T[x,a,b]*T[j,c,i]-1/4*T[a,x,j]*T[c,b,i]+1/4*d[x](T[a,b,c],[X])+1/4*d[x](T[b,a,c],[X])-1/2*d[x](T[c,a,b],[X])-1/4*d[a](T[b,x,c],[X])+1/2*d[b](T[x,a,c],[X])-1/4*d[b](T[a,x,c],[X]);
```

$$e1 := \frac{\left(2\,T_{x,j,g}\,T_{i,c,k}-2\,T_{x,j,c}\,T_{i,g,k}-2\,d_{j}\!\left(T_{x,c,i},[X]\right)+2\,d_{x}\!\left(T_{c,j,i},[X]\right)\right)g_{-a,b}}{4} + \frac{\left(T_{a,x,b}+T_{b,x,a}\right)T_{c,j,i}}{4} - \frac{T_{b,x,j}\,T_{c,a,i}}{4} - \frac{T_{x,j,a}\,T_{i,b,c}}{2} + \frac{T_{x,a,b}\,T_{j,c,i}}{2} - \frac{T_{a,x,j}\,T_{c,b,i}}{4} + \frac{d_{x}\!\left(T_{a,b,c},[X]\right)}{4} + \frac{d_{x}\!\left(T_{b,a,c},[X]\right)}{4} - \frac{d_{x}\!\left(T_{c,a,b},[X]\right)}{2} - \frac{d_{a}\!\left(T_{b,x,c},[X]\right)}{4} + \frac{d_{b}\!\left(T_{x,a,c},[X]\right)}{2} - \frac{d_{b}\!\left(T_{a,x,c},[X]\right)}{4}$$

```
> e2 := 1/4*(2*T[x,j,g]*T[i,c,k]-2*T[x,j,c]*T[i,g,k]+2*d[j](T[i,x,c],[X])-2*d[x](T[j,c,i],[X]))*g_[a,b]+1/4*(T[a,x,b]+T[b,x,a])*T[c,j,i]-1/4*T[b,x,j]*T[c,a,i]-1/2*T[x,j,a]*T[i,b,c]+1/2*T[x,a,b]*T[j,c,i]-1/4*T[a,x,j]*T[c,b,i]+1/4*d[x](T[a,b,c],[X])+1/4*d[x](T[b,a,c],[X])-1/2*d[x](T[c,a,b],[X])-1/4*d[a](T[b,x,c],[X])+1/2*d[b](T[x,a,c],[X])-1/4*d[b](T[a,x,c],[X]);
```

$$e2 := \frac{\left(2\,T_{x,j,g}\,T_{i,c,k}-2\,T_{x,j,c}\,T_{i,g,k}+2\,d_{j}\!\left(T_{i,x,c},[X]\right)-2\,d_{x}\!\left(T_{j,c,i},[X]\right)\right)g_{-a,b}}{4} + \frac{\left(T_{a,x,b}+T_{b,x,a}\right)T_{c,j,i}}{4} - \frac{T_{b,x,j}\,T_{c,a,i}}{4} - \frac{T_{x,j,a}\,T_{i,b,c}}{2} + \frac{T_{x,a,b}\,T_{j,c,i}}{2} - \frac{T_{a,x,j}\,T_{c,b,i}}{4} + \frac{d_{x}\!\left(T_{a,b,c},[X]\right)}{4} + \frac{d_{x}\!\left(T_{b,a,c},[X]\right)}{4} - \frac{d_{x}\!\left(T_{c,a,b},[X]\right)}{2} - \frac{d_{a}\!\left(T_{b,x,c},[X]\right)}{4}$$

$$+\frac{d_{b}\left(T_{x,a,c}, [X]\right)}{2}-\frac{d_{b}\left(T_{a,x,c}, [X]\right)}{4}$$

```
> evalb(e1=e2);
```

$$false$$

- The new package ExpressionTools permits you to perform a visual comparison of two expressions.
- The main command in this package is Compare, which, given two expressions, compares them and highlights their differences:

```
> with(ExpressionTools);
```

$$[Compare, Options]$$

```
> Compare(e1, e2);
```

$$\frac{\left(2\,T_{x,j,g}\,T_{i,c,k}-2\,T_{x,j,c}\,T_{i,g,k}-2\,d_{\begin{pmatrix} j \\ x \end{pmatrix}}\left(T_{\begin{pmatrix} x \\ j \end{pmatrix},c,i}, [X]\right)+2\,d_{\begin{pmatrix} x \\ j \end{pmatrix}}\left(T_{\begin{pmatrix} c \\ i \end{pmatrix},\begin{pmatrix} j \\ x \end{pmatrix},\begin{pmatrix} i \\ c \end{pmatrix}}, [X]\right)\right)g_{-a,b}}{4}$$
$$+\frac{\left(T_{a,x,b}+T_{b,x,a}\right)T_{c,j,i}}{4}-\frac{T_{b,x,j}\,T_{c,a,i}}{4}-\frac{T_{x,j,a}\,T_{i,b,c}}{2}+\frac{T_{x,a,b}\,T_{j,c,i}}{2}-\frac{T_{a,x,j}\,T_{c,b,i}}{4}$$
$$+\frac{d_{x}\left(T_{a,b,c}, [X]\right)}{4}+\frac{d_{x}\left(T_{b,a,c}, [X]\right)}{4}-\frac{d_{x}\left(T_{c,a,b}, [X]\right)}{2}-\frac{d_{a}\left(T_{b,x,c}, [X]\right)}{4}$$
$$+\frac{d_{b}\left(T_{x,a,c}, [X]\right)}{2}-\frac{d_{b}\left(T_{a,x,c}, [X]\right)}{4}$$

In the above example, the two expressions are combined into one. Differing subexpressions are replaced by a vector containing the two corresponding subexpressions, each highlighted in a different color to indicate which of the original expressions it was a part of.

- For smaller expressions (both having length less than 250, an adjustable parameter), the two expressions are printed one after the other, and the differences are simply highlighted by changing their colors:

```
> x1 := (a-b)*(A-B)*c;
```

$$x1 := (a-b)\,(A-B)\,c$$

```
> x2 := (b-a)*(B-A)*C;
```

$$x2 := (b-a)\,(B-A)\,C$$

```
> Compare(x1, x2);
```

$$(a + -1\,b)\;(A + -1\,B)\;c$$
$$(b + -1\,a)\;(B + -1\,A)\;C$$

- If a verification is provided as an option, the comparison ignores any differences it can find that satisfy that verification:

```
> Compare(x1, x2, sign);
```

$$(a - b)\;(A - B)\;c$$
$$(b - a)\;(B - A)\;C$$

- For sums, products, and sets, operands are compared according to the best match, even when none of the top-level subexpressions match exactly:

```
> x3, x4 := a/2+b/3+c/4, c/5+b/6+d/7:
```

```
> Compare(x3, x4, combine);
```

$$\begin{pmatrix} \frac{a}{2} \\ \frac{d}{7} \end{pmatrix} + \begin{pmatrix} \frac{1}{3} \\ \frac{1}{6} \end{pmatrix} b + \begin{pmatrix} \frac{1}{4} \\ \frac{1}{5} \end{pmatrix} c$$

Note in the above that we used the combine option, which forces display in combined format even though the expressions were small.

- Here's a larger example, showing the use of some other options to control the display:

```
> x3 := Vector(2,[w(x), z(x)]) = Vector(2,[1/15*exp(-x)*(-15*c__1+3)
  +1/15*(10*c__2+2)*exp(4*x)-1/3*exp(x), 1/5*exp(-x)*(-1+5*c__1)
  +1/5*exp(4*x)*(1+5*c__2)]);
```

$$x3 := \begin{bmatrix} w(x) \\ z(x) \end{bmatrix} = \begin{bmatrix} \frac{e^{-x}\left(-15\,c_1 + 3\right)}{15} + \frac{\left(10\,c_2 + 2\right) e^{4\,x}}{15} - \frac{e^{x}}{3} \\ \frac{e^{-x}\left(-1 + 5\,c_1\right)}{5} + \frac{e^{4\,x}\left(1 + 5\,c_2\right)}{5} \end{bmatrix}$$

```
> x4 := Vector(2,[w(x), z(x)]) = Vector(2, [1/15*exp(-x)*(-15*_C1+3)
  +1/15*(10*_C2+2)*exp(4*x)-1/3*exp(x),1/5*exp(-x)*((1+5*_C2)*exp(5*
  x)-1+5*_C1)]);
```

$$x4 := \begin{bmatrix} w(x) \\ z(x) \end{bmatrix} = \begin{bmatrix} \frac{e^{-x}(-15\_C1+3)}{15} + \frac{(10\_C2+2)e^{4x}}{15} - \frac{e^{x}}{3} \\ \frac{e^{-x}((1+5\_C2)e^{5x} - 1 + 5\_C1)}{5} \end{bmatrix}$$

```
> Compare(x3, x4, Silver, Gold, showvectorbrackets=false,
  showNULLoperands);
```

$$\begin{bmatrix} w(x) \\ z(x) \end{bmatrix} =$$

$$\begin{bmatrix} \frac{e^{-x}\left(-15 \begin{matrix} c_1 \\ \_C1 \end{matrix} + 3\right)}{15} + \frac{\left(10 \begin{matrix} c_2 \\ \_C2 \end{matrix} + 2\right)e^{4x}}{15} - \frac{e^{x}}{3} \\ \frac{e^{-x}(-1+5c_1)}{5} + \frac{e^{\begin{matrix} 4 \\ -1 \end{matrix} x}\left(\left(1 + 5\_C2\right) e^{5x} + 5 \begin{matrix} c_2 \\ \_C1 \end{matrix} + -1\right)}{5} \end{bmatrix}$$

- Compare uses the uneval parameter modifier, which means it does not evaluate its arguments before comparing them:

```
> x5 := 'hypergeom'([a,b],[c],x);
```

$$x5 := \mathrm{hypergeom}([a, b], [c], x)$$

```
> x6 := 'hypergeom'([b,a],[c],x);
```

$$x6 := \mathrm{hypergeom}([b, a], [c], x)$$

```
> evalb(x5=x6);
```

$$true$$

```
> Compare(x5, x6);
```

$$\mathrm{hypergeom}([a, b], [c], x)$$
$$\mathrm{hypergeom}([b, a], [c], x)$$

- Note, however, that this could lead to unexpected results:

```
> Compare(x1, subs(C=c, x2), combine, sign);
```

$$\left(\begin{array}{c}(a-b)(A-B)\,c\\ subs(C=c,x2)\end{array}\right)$$

- So be sure to evaluate results if desired before passing them to Compare:

```
> x3 := subs(C=c, x2);
```

$$x3 := (b-a)\,(B-A)\,c$$

```
> Compare(x1, x3, combine, sign);
```

The expressions satisfy the given verification

Or alternatively you can use the evaluate option to force evaluation:

```
> Compare(x5, x6, evaluate);
```

The expressions are the same

```
> Compare(x1, subs(C=c, x2), sign, evaluate);
```

The expressions satisfy the given verification

- A secondary ExpressionTools command called Options allows you to query and set the default values for each option available to the Compare command:

```
> Options();
```

backgroundcolor : [LightGreen, Pink]

combine : default

combinethreshold : 250

evaluate : false

foregroundcolor : [Green, Red]

highlight : default

showNULLoperands : false

showvectorbrackets : true

verification : boolean

```
> x1 := (2*a-3*b)*(A-B)*c:
```

```
> x2 := (3*b-2*a)*(B-A)*C:
```

```
> Compare(x1, x2);
```

$$(2\,a+-3\,b)\,(A+-1\,B)\,c$$
$$(3\,b+-2\,a)\,(B+-1\,A)\,C$$

```
> Options(combine, highlight, backgroundcolor);
```

combine : default

highlight : default

backgroundcolor : [LightGreen, Pink]

```
> Options(combine=true, highlight=background, backgroundcolor =
  [Yellow, Orange]);
```

combine : true

highlight : background

backgroundcolor : [Yellow, Orange]

```
> Compare(x1, x2);
```

$$\left(\begin{pmatrix} 2 \\ -2 \end{pmatrix} a + \begin{pmatrix} -3 \\ 3 \end{pmatrix} b\right) \left((-1)\, A + (-1)\, B\right) \begin{pmatrix} c \\ C \end{pmatrix}$$